



APPLICATIONS LIBRARY PROGRAM

TITLE Leroy Character Generator		ABSTRACT NUMBER 51/00-8012/0
ORIGINAL DATE 6/77	REVISION DATE	MEMORY REQUIREMENT 9705 bytes (full version)
AUTHOR Mark Mehall		PERIPHERALS 4662 (Optional)

ABSTRACT

This set of routines draws alphanumeric characters and special symbols on a 4051 or 4662. The characters and symbols can be any size, at any angle and may be positioned at any location on the screen or plotter surface. Since the routines do not interfere with the user's window and viewport, they can be used with existing programs for annotation. Labels can be placed on axes, points can be indicated, and legends can be added to graphs. There are 128 possible symbols, including upper and lower-case letters, numbers, sixteen centered symbols, and special characters. The routines also offer subscripts and superscripts, underlining, *italics*, tab settings, backspace and carriage return functions. The characters are similar to the Leroy style.

The program material contained herein is supplied without warranty or representation of any kind. Tektronix, Inc., assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

DESCRIPTION

There are three routines in this package and one data file. The first routine (GOSUB 9000) initializes the tables used by the program from the data file. This routine should be used only once during each program. The second routine (GOSUB 9100) does the actual character generation. The routine requires three variables: C\$ - the string of ASCII characters to be drawn, S - the size of the characters in Graphic Display Units (GDUs), and A - the angle (in degrees) of counterclockwise rotation. The third routine (GOSUB 9200) is an entry point to allow some special features to be used. It will be discussed later.

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

OPERATING INSTRUCTIONS

Insert the tape into the 4051 and load the program into memory. The user then must enter the parameter program to initialize the tables, set the angle, size the characters, define the character string, determine the X, Y point to begin the draw, followed with a GOSUB 9100 and END (see the 3 examples attached).

SAVING PARAMETER PROGRAMS

Parameter programs saved on a separate file must include a dummy 9000 REM statement. To run these with the Leroy Character Generator, load the parameter program FIND the # of the Leroy program and APPEND 9000.

LIMITATIONS

The Leroy program is coded to run on the 4051 Graphic Display Screen. If you are writing to the plotter, change line 9010 to read 9010 U9=1.

The explicit instructions for using the many features of this program follow the page of characters available (page 9).

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

EXAMPLE #1 PROGRAM

```
100 REM INITIALIZE THE CHARACTER GENERATOR
110 GOSUB 9000
120 REM SET ANGLE TO 0
130 A=0
140 REM SET CHARACTER STRING
150 C$="A Character String"
160 REM SET THE SCALE TO 4 (GDUs)
170 S=4
180 REM MOVE TO THE BEGINNING LOCATION
190 MOVE 65,50
200 REM EXECUTE THE ROUTINE
210 GOSUB 9100
220 END
```

EXAMPLE #1 RESULT

A Character String

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

EXAMPLE #3 PROGRAM

```
100 REM SET CHARACTER STRING
110 C$=" Symbol Generator"
120 REM SET THE SCALE TO 2.5 (GDUs)
130 S=2.5
140 REM SET THE ANGLE
150 FOR A=0 TO 330 STEP 30
160 REM MOVE TO THE BEGINNING LOCATION
170 MOVE 65,50
180 REM EXECUTE THE ROUTINE
190 GOSUB 9100
200 NEXT A
210 END
```

Leroy Character Generator

51/00-8012/0

EXAMPLE #3 RESULT

A circular arrangement of the text "Symbol Generator" repeated multiple times, creating a star-like pattern. The text is written in a simple, sans-serif font and is oriented radially, with each instance of the words pointing towards the center of the circle. The repetition of the text creates a complex, overlapping geometric design.

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

CHARACTERS AVAILABLE

There are 128 characters in this set of routines. The printable ASCII characters (decimal 32 through 126) may be obtained directly from the 4051 keyboard. The ASCII control codes (decimal 0 through 31) have been defined to produce special characters (see table). Decimal 0 through 15 are centered symbols and decimal 16 through 31 are mathematical characters. The control code characters can be obtained by using the CTRL key on the 4051. For example C\$="A" (CTRL A) produces ⓐ.

There are two characters which cannot be obtained directly from the keyboard. Both decimal 13 and 127 must be defined using the CHR function (e.g. C\$=CHR(127)).

TITLE

LEROY CHARACTER GENERATOR

51/00-8012/0

Characters Available

0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
□	○	△	◇	×	+	▽	⊗	⊙	⊕	⊗	⊗	⊕	⊕	⊕	⊕
16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31
°	↖	↗	↘	↙	↘	↗	↖	↗	↖	↗	↖	↗	↖	↗	↖
32	33	34	35	36	37	38	39	40	41	42	43	44	45	46	47
!	!"	#	\$	%	&	'	()	*	+	,	-	.	/	
48	49	50	51	52	53	54	55	56	57	58	59	60	61	62	63
0	1	2	3	4	5	6	7	8	9	:	;	<	=	>	?
64	65	66	67	68	69	70	71	72	73	74	75	76	77	78	79
@	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
80	81	82	83	84	85	86	87	88	89	90	91	92	93	94	95
P	Q	R	S	T	U	V	W	X	Y	Z	[\]	^	_
96	97	98	99	100	101	102	103	104	105	106	107	108	109	110	111
'	a	b	c	d	e	f	g	h	i	j	k	l	m	n	o
112	113	114	115	116	117	118	119	120	121	122	123	124	125	126	127
p	q	r	s	t	u	v	w	x	y	z	{	}	~	▽	

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

FEATURES

The control character | is used to convey special instructions according to the character which follows the |. For example, |7 means to enter the subscript mode. If a | is followed by a character which is not defined as a code, it will be ignored. A single | is drawn if two of them appear successively. Control characters may appear anywhere in C\$.

Superscripts and subscripts

If the characters |5, |6, |7, or |8, are encountered, an instruction regarding superscripts and subscripts is executed:

- |5 means to enter the superscript mode;
- |6 means to leave the superscript mode;
- |7 means to enter the subscript mode;
- |8 means to leave the subscript mode.

Superscripts may have subscripts or superscripts, etc. For example the string C\$="P|52|6Q" produces the result:

$$P^2Q$$

and the string C\$="e|5x|7k|8|52(n-1)|6|6dx|7k|8" produces the result:

$$e^{x_k^{2(n-1)}} dx_k$$

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

Establishing a reference point and returning to it

If the characters |9 are encountered, the coordinates of the next available position are stored for future reference.

If the characters |A are encountered, the drawing will continue at the position stored in the last reference to |9.

If |9 has not used, the reference point is assumed to be the starting location used when GOSUB 9100 was invoked. Since GOSUB 9100 resets the reference point, GOSUB 9200 should be used with |9.

Tab

A single "tab setting" is provided. The position of the tab setting is established (or revised) by using the control characters |B in C\$ when at the desired location. Any subsequent use of the control characters |C will cause the following characters to be drawn at a point directly below the point at which |B was invoked. If |B has not been used, it is assumed that the tab was set at the starting location when GOSUB 9100 was invoked. The tab (|C) is usually used with multiple calls to carriage return (see below). The "tab set" position is preserved from one call to GOSUB 9100 or GOSUB 9200 to another.

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

Underlining

The underscoring of a word or a series of words is accomplished by the use of the code |9 to establish the beginning of the underscored material and the code |D to establish the end of the underscored material. For example the string C\$="The |9title|D" produces the result:

The title

If |D is invoked without |9 ever having been used, underlining will start from the starting location when GOSUB 9100 was executed.

Backspace

If the characters |E are encountered, drawing will continue *starting at the beginning of the last drawn character.*

Backspace may not be called meaningfully more than once in succession, since the only pertinent information stored is the beginning location of the last plotted character.

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

Carriage return

The control characters |F are used when drawing is to be resumed at a point located at $1.5 \times S$ below the beginning point of the previous execution of GOSUB 9100. Carriage returns may be used successively.

Italics

When a group of letters is to be *italicized*, the control characters |G are placed before the group and the control characters |H after the group, in C\$. For example, the string C\$="|Gtime|H, T" produces the result:

time, T

Italics are "carried over" from one call to GOSUB 9100 (or GOSUB 9200) to the next.

For example, the series of statements:

```
C$="|G"
```

```
GOSUB 9100
```

```
C$=STR(PI)
```

```
GOSUB 9200
```

```
C$="|H r|52"
```

```
GOSUB 9200
```

produces the result:

3.14159265359 r²

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

MEMORY REQUIREMENTS

The full version takes 9705 bytes of storage. The following can be used to decrease the amount of memory required:

1. Delete all REM statements (except 9000, 9100 and 9200). This version requires 6683 bytes.
2. Remove the special functions by deleting lines 9226 through 9364. This version uses 5334 bytes.

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9000 REM THIS ROUTINE INITIALIZES THE TABLES USE BY THE
9002 REM CHARACTER GENERATOR. IT SHOULD BE EXECUTED (ONCE) BEFORE
9004 REM THE CHARACTER GENERATOR IS USED. (i.e. GOSUB 9000)

9006 DIM N$(257),D$(513),M$(2279),E$(1),F$(1),G$(1)

9008 REM SET GRAPHIC DEVICE (DEFAULT IS SCREEN)

9010 U9=32
9011 PRINT "FILE # OF DATA? ";
9012 INPUT Z
9013 PAGE
9014 FIND Z
9016 READ @33:N$,D$,M$

9018 REM SET CODE FOR NO ITALICS

9020 I9=0
9022 RETURN

9100 REM ***CHARACTER GENERATOR ROUTINE*** 51/00-8012/0
9102 REM VARIABLES USED -
9104 REM S - SCALE OF THE CHARACTERS IN GRAPHIC DISPLAY UNITS (GDUs)
9106 REM C$ - ASCII CHARACTER STRING TO BE PLOTTED
9108 REM A - ANGLE (IN DEGREES) OF COUNTERCLOCKWISE ROTATION
9109 REM Z - DATA FILE #
9110 REM LOCAL VARIABLES USED -
9112 REM X4,Y4 - TAB REFERENCE
9114 REM X5,Y5 - GENERAL REFERENCE POINT ALSO UNDERLINE REFERENCE
9116 REM X6,Y6 - CURRENT LOCATION
9118 REM X7,Y7 - CHARACTER MOVES
9120 REM X8,Y8 - STARTING CHARACTER STRING LOCATION USED FOR CR
9122 REM X9,Y9 - WORK VARIABLES
9124 REM S9,C9 - SIN(A) AND COS(A)
9126 REM E$,F$,G$ - WORK STRINGS
```

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9128 REM  N$ - TABLE OF MOVES PER CHARACTER
9130 REM  D$ - TABLE OF INDEX INTO M$ AND WIDTH
9132 REM  M$ - CHARACTER MOVES
9134 REM  I9 - ITALICS CODE (1=ON 0=OFF)
9136 REM  L9,M9 - INDEX VARIABLES
9138 REM  P9 - DRAW CODE (1=MOVE 2=DRAW)
9140 REM  S8 - CORRECTED SCALE FACTOR
9142 REM  U9 - UNIT NUMBER OF GRAPHIC DEVICE (DEFAULT IS 32)
9144 REM  C8 - ASCII DECIMAL EQUIVALENT OF LETTER OF C$
9146 REM  W8 - CHARACTER WIDTH (IN GDUs) USED FOR BACKSPACE
9148 REM  N9,D9 - MISC. VARIABLES
9150 REM GET BEGINNING LOCATION

9152 INPUT @U9,24:X6,Y6

9154 REM SAVE STARTING LOCATION

9156 X8=X6
9158 Y8=Y6

9160 REM SET REFERENCE POINT

9162 X5=X6
9164 Y5=Y6

9200 REM ENTRY POINT FOR BYPASSING THE GIN COMMAND
9202 REM THE CHARACTERS WILL BE DRAWN AT THE END OF THE LAST STRING
9204 REM CORRECT SCALE FACTOR FOR GDUs

9206 S8=S*1.25
9208 SET DEGREES

9210 REM ALL ANGLES IN DEGREES

9212 S9=SIN(A)
9214 C9=COS(A)
```


TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9216 FOR L9=1 TO LEN(C$)
9218     REM GET A CHARACTER
9220     E$=SEG(C$,L9,1)
9222     C8=ASC(E$)
9224     REM CHECK FOR CONTROL CHARACTER
9226     IF C8<>124 THEN 9366
9228     L9=L9+1
9230     E$=SEG(C$,L9,1)
9232     C8=ASC(E$)
9234     REM CHECK FOR ANOTHER CONTROL CHARACTER, IF YES DRAW ONE
9236     IF C8=124 THEN 9366
9238     REM CHECK FOR 5 THROUGH 9
9240     IF C8<53 OR C8>57 THEN 9246
9242     GO TO C8-52 OF 9252,9262,9272,9282,9292
9244     REM CHECK FOR A THROUGH G
9246     IF C8<65 OR C8>72 THEN 9520
9248     GO TO C8-64 OF 9300,9308,9316,9324,9332,9340,9356,9362
9250     REM ENTER SUPERScript MODE (CODE 5)
9252     Y6=Y6+S8*0.3333*C9
9254     X6=X6-S8*0.3333*S9
9256     S8=S8*0.6
9258     GO TO 9518
9260     REM LEAVE SUPERScript MODE (CODE 6)
```

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9262      S8=S8*1.6667
9264      Y6=Y6-S8*0.3333*C9
9266      X6=X6+S8*0.3333*S9
9268      GO TO 9518

9270      REM ENTER SUBSCRIPT MODE (CODE 7)

9272      Y6=Y6-S8*0.2667*C9
9274      X6=X6+S8*0.2667*S9
9276      S8=S8*0.6
9278      GO TO 9518

9280      REM LEAVE SUBSCRIPT MODE (CODE 8)

9282      S8=S8*1.6667
9284      Y6=Y6+S8*0.2667*C9
9286      X6=X6-S8*0.2667*S9
9288      GO TO 9518

9290      REM ESTABLISH REFERENCE POINT (CODE 9)

9292      X5=X6
9294      Y5=Y6
9296      GO TO 9520

9298      REM RETURN TO REFERENCE POINT (CODE A)

9300      X6=X5
9302      Y6=Y5
9304      GO TO 9518

9306      REM ESTABLISH TAB REFERENCE (CODE B)

9308      X4=X6-X8
9310      Y4=Y6-Y8
```

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9312      GO TO 9520

9314      REM TAB OPERATOR (CODE C)

9316      X6=X4+X8
9318      Y6=Y4+Y8
9320      GO TO 9518

9322      REM UNDERLINE (CODE D)

9324      PRINT @U9,21:X6+0.2667*S8*S9,Y6-0.2667*S8*C9
9326      PRINT @U9,20:X5+0.2667*S8*S9,Y5-0.2667*S8*C9
9328      GO TO 9518

9330      REM BACKSPACE (CODE E)

9332      X6=X6-X9*C9
9334      Y6=Y6-X9*S9
9336      GO TO 9518

9338      REM CARRIAGE RETURN (CODE F)

9340      X9=X8+1.5*S8*S9
9342      Y9=Y8-1.5*S8*C9
9344      X6=X9
9346      Y6=Y9
9348      X8=X9
9350      Y8=Y9
9352      GO TO 9518

9354      REM ITALICS MODE (CODE G)

9356      I9=1
9358      GO TO 9520

9360      REM NO ITALICS (CODE H)
```

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9362      I9=0
9364      GO TO 9520
9366      P9=2
9368      N9=C8*2+1
9370      E$=SEG(N$,N9,1)
9372      F$=SEG(N$,N9+1,1)
9374      N9=ASC(E$)*16+ASC(F$)
9376      D9=C8*4+1
9378      E$=SEG(D$,D9,1)
9380      F$=SEG(D$,D9+1,1)
9382      G$=SEG(D$,D9+2,1)
9384      W8=D9
9386      D9=ASC(E$)*256+ASC(F$)*16+ASC(G$)
9388      D9=D9*2+1
9390      E$=SEG(D$,W8+3,1)
9392      W8=ASC(E$)
9394      FOR M9=D9 TO D9+2*N9-1 STEP 2
9396          X$=SEG(M$,M9,1)
9398          X7=ASC(X$)
9400          Y$=SEG(M$,M9+1,1)
9402          Y7=ASC(Y$)

9404      REM CHECK FOR SPECIAL CODE

9406      IF X7=15 THEN 9474

9408      REM OFFSET FOR NONCENTERED SYMBOLS

9410      X7=X7-1
9412      Y7=Y7-3

9414      REM CHECK FOR CENTERED SYMBOLS

9416      IF C8>16 THEN 9426
```

TITLE

ABSTRACT NO:

LEROY CHARACTER GENERATOR

51/00-8012/0

9418 REM OFFSET FOR CENTERED SYMBOLS

9420 Y7=Y7-6

9422 X7=X7-6

9424 REM ITALICIZE IT

9426 IF I9=0 THEN 9432

9428 X7=X7+0.3*Y7

9430 REM ROTATE IT

9432 IF A=0 THEN 9444

9434 X9=-Y7*S9+X7*C9

9436 Y9=Y7*C9+X7*S9

9438 X7=X9

9440 Y7=Y9

9442 REM SCALE IT

9444 X9=X7*0.06667*S8+X6

9446 Y9=Y7*0.06667*S8+Y6

9448 X7=X9

9450 Y7=Y9

9452 REM CHECK FOR A MOVE

9454 IF P9=2 THEN 9464

9456 REM DRAW IT

9458 PRINT @U9,20:X7,Y7

9460 GO TO 9468

9462 REM MOVE TO IT

TITLE

LEROY CHARACTER GENERATOR

ABSTRACT NO:

51/00-8012/0

```
9464      PRINT @U9,21:X7,Y7

9466      REM RESET DRAW CODE

9468      P9=1
9470      GO TO 9500

9472      REM SPEC CODE X7=15

9474      GO TO Y7+1 OF 9482,9498,9490
9476      GO TO 9500

9478      REM MOVE CODE
9480      REM Y7=0

9482      P9=2
9484      GO TO 9500

9486      REM Y7=2
9488      REM INCREASE SCALE FACTOR

9490      S8=S8*2
9492      GO TO 9500

9494      REM Y7=1
9496      REM DECREASE SCALE FACTOR

9498      S8=S8*0.5
9500      NEXT M9

9502      REM CHECK FOR CENTERED SYMBOL (RETURN TO CENTER OF SYMBOL)

9504      IF C8<17 THEN 9518

9506      REM CALCULATE WIDTH OF CHARACTER
```

TITLE

LEROY CHARACTER GENERATOR

51/00-8012/0

```
9508      X9=W8*0.06667*S8
9510      REM CALCULATE NEXT LOCATION
9512      Y6=Y6+X9*S9
9514      X6=X6+X9*C9
9516      REM MOVE TO IT
9518      PRINT @U9,21:X6,Y6
9520 NEXT L9
9522 RETURN
```

TITLE

Leroy Character Generator

ABSTRACT NO:

51/00-8012/0

REMARKS

This set of routines is based on a program developed at the University of Illinois by J. W. Phillips. Please send any comments or suggestions to:

Mark Mehall
Tektronix, Inc.
5350 Keystone Court
Rolling Meadows, Ill. 60008